



# DECUS

## PROGRAM LIBRARY

DECUS NO.	8-462
TITLE	INSTIN
AUTHOR	Paul Kinzelman
COMPANY	Carnegie-Mellon University Pittsburgh, Pennsylvania
DATE	August 11, 1971
SOURCE LANGUAGE	BASIC

### ATTENTION

This is a USER program. Other than requiring that it conform to submittal and review standards, no quality control has been imposed upon this program by DECUS.

The DECUS Program Library is a clearing house only; it does not generate or test programs. No warranty, express or implied, is made by the contributor, Digital Equipment Computer Users Society or Digital Equipment Corporation as to the accuracy or functioning of the program or related material, and no responsibility is assumed by these parties in connection therewith.

20330

1954



THE UNIVERSITY OF CHICAGO

LIBRARY



# INSTIN

DECUS Program Library Write-up

DECUS NO. 8-462

## Description of Program:

INSTIN is a program which will solve Instant Insanity. The puzzle consists of four cubes, each side of which is colored white, red, green, or blue. To solve the puzzle, one must stack the cubes in a line so that each color appears only once along a side which is four cubes long. The program will find and print out all the basic solutions. The program allows the user to change the puzzle by switching colored sides or by changing the color of sides. The user may inhibit the printing of the solutions to determine the number of basic solutions quickly.

## Operation:

The program should be self-explanatory.





```

10 DIM Z(4),C(4,6),D(6)
50 PRINT "PROGRAM FOR SOLVING INSTANT INSANITY"
53 REM ...BY PAUL M. KINZELMAN
55 REM 1415 OTIS ST. N.E.
57 REM WASH. D.C. 20017
59 REM DEC 1970, REVISED JUNE 1971
60 PRINT "EXPLAIN(1=YES,0=NO)";
61 INPUT D
62 IF D=0 GOTO 130
69 PRINT "REPRESENTATION-"
70 PRINT "WHITE=1 RED=2 GREEN=3 BLUE=4"
80 PRINT "STACK SO THAT TOPS AND BOTTOMS ARE HIDDEN:"
81 PRINT
82 PRINT "STARTING POINT FORMAT:"
83 PRINT " BACK,LEFT,FRONT,RIGHT, TOP,BOTTOM"
84 PRINT
85 PRINT
86 PRINT "OUTPUT FORMAT:"
87 PRINT
90 PRINT " TOP"
100 PRINT "BACK LEFT FRONT RIGHT"
110 PRINT " BOTTOM"
120 PRINT
130 PRINT
170 FOR I=1 TO 4
180 PRINT "CUBE";I;"=";
200 INPUT C(I,1),C(I,2),C(I,3),C(I,4),C(I,5),C(I,6)
220 PRINT
221 NEXT I
223 LET S=0
225 PRINT "PRINT SOLUTIONS(1=YES,0=NO)";
227 INPUT D
232 LET N=1
234 FOR J=1 TO 4
236 LET Z(J)=1
238 NEXT J
240 LET N=N+1
250 REM FUNCTION TO CHECK IF STACK IS OK FFFFFFFFFFFFFFFFFFFFFFFFFF
260 FOR I=1 TO 4
270 FOR J=2 TO N
280 FOR K=J TO N
290 IF C(J-1,I)=C(K,I) GOTO 350
300 NEXT K
310 NEXT J
320 NEXT I
325 REM END OF FUNCTION FFFFFFFFFFFFFFFFFFFFFFFFFF
330 IF N=4 GOTO 614
340 GOTO 240
345 REM SUBROUTINE TO CHANGE CUBE SSSSSSSSSSSSSSSSSSSSS
350 FOR I=1 TO 6
360 LET D(I)=C(N,I)
370 NEXT I
375 IF N=1 GOTO 505
380 IF INT ( Z(N)/12)*12= Z(N) GOTO 490
390 IF INT(Z(N)/4)*4=Z(N) GOTO 440
400 FOR I=1 TO 3
410 LET C(N,5-I)=C(N,4-I)
415 NEXT I
420 LET C(N,1)=D(4)
430 GOTO 530
440 LET C(N,2)=D(6)

```





```

450 LET C(N,4)=D(5)
460 LET C(N,5)=D(2)
470 LET C(N,6)=D(4)
480 GOTO 530
490 LET C(N,1)=D(3)
500 LET C(N,3)=D(1)
501 LET C(N,5)=D(6)
502 LET C(N,6)=D(5)
503 GOTO 530
505 LET Z(N)=Z(N)+7
507 FOR M=1 TO 3
509 LET C(N,M)=C(N,M+1)
511 NEXT M
513 LET C(N,4)=D(1)
515 FOR I=1 TO 4
517 LET D(I)=C(N,I)
519 NEXT I
521 GOTO 380
525 REM END OF SUBROUTINE SSSSSSSSSSSSSSSSSSS
530 IF Z(N)<>24 GOTO 580
540 IF N=1 GOTO 600
550 LET Z(N)=1
560 LET N=N-1
570 GOTO 350
580 LET Z(N)=Z(N)+1
590 GOTO 260
600 PRINT "ALL POSSIBILITIES TESTED. NUMBER OF SOLN'S =";S
610 GOTO 720
614 LET S=S+1
617 IF D=0 GOTO 350
620 PRINT "-----"
630 PRINT "ONE SOLUTION:"
640 FOR I=1 TO 4
650 PRINT
660 PRINT "          ";C(I,5)
670 PRINT C(I,1);C(I,2);C(I,3);C(I,4)
680 PRINT "          ";C(I,6)
685 NEXT I
690 PRINT
700 PRINT "-----"
710 GOTO 350
720 PRINT
721 PRINT "PRINT CURRENT POSITION(1=YES,0=NO)";
722 INPUT I
723 IF I=0 GOTO 730
724 FOR I=1 TO 4
725 PRINT
726 PRINT "          ";C(I,5)
727 PRINT C(I,1);C(I,2);C(I,3);C(I,4)
728 PRINT "          ";C(I,6)
729 NEXT I
730 PRINT
731 FOR I=1 TO 4
740 PRINT "DO YOU WISH TO CHANGE CUBE";I;"(1=YES,0=NO)";
750 INPUT X
760 IF X=0 GOTO 790
770 PRINT "CUBE";I;"=";
780 INPUT C(I,1),C(I,2),C(I,3),C(I,4),C(I,5),C(I,6)
790 NEXT I
800 GOTO 223
810 END

```





RUN

PROGRAM FOR SOLVING INSTANT INSANITY

EXPLAIN(1=YES,0=NO)? 1

REPRESENTATION-

WHITE=1      RED=2      GREEN=3      BLUE=4

STACK SO THAT TOPS AND BOTTOMS ARE HIDDEN:

STARTING POINT FORMAT:

BACK, LEFT, FRONT, RIGHT, TOP, BOTTOM

OUTPUT FORMAT:

		TOP		
BACK	LEFT	FRONT	RIGHT	
		BOTTOM		

CUBE 1 =? 1, 1, 3, 2, 4, 2

CUBE 2 =? 2, 2, 2, 3, 4, 1

CUBE 3 =? 1, 1, 4, 3, 3, 2

CUBE 4 =? 4, 4, 2, 1, 3, 3

PRINT SOLUTIONS(1=YES,0=NO)? 1

-----  
ONE SOLUTION:

		2		
2	1	4	3	
		1		
		2		
3	4	2	1	
		2		
		1		
1	3	3	2	
		4		
		3		
4	2	1	4	
		3		





ALL POSSIBILITIES TESTED. NUMBER OF SOLN'S = 1

PRINT CURRENT POSITION(1=YES,0=NO)? 1

1  
3 4 1 2  
2

4  
2 2 2 3  
1

3  
1 1 4 3  
2

3  
4 4 2 1  
3

DO YOU WISH TO CHANGE CUBE 1 (1=YES,0=NO)? 0

DO YOU WISH TO CHANGE CUBE 2 (1=YES,0=NO)? 1

CUBE 2 =? 2,2,3,1,2,4

DO YOU WISH TO CHANGE CUBE 3 (1=YES,0=NO)? 00

DO YOU WISH TO CHANGE CUBE 4 (1=YES,0=NO)? 0

PRINT SOLUTIONS(1=YES,0=NO)? 1

-----  
ONE SOLUTION:

1  
3 4 1 2  
2

3  
2 2 4 1  
2

2  
1 1 3 4  
3

1  
4 3 2 3  
4

-----  
ONE SOLUTION:

1  
3 4 1 2  
2

3  
2 2 4 1  
2

2  
1 1 3 4  
3

4  
4 3 2 3  
1

